P5

PRÁCTICA-REPRESENTACIÓN DE CURVAS EN COORDENADAS POLARES

▼ Ejercicio Propuesto P-5.1

Dibujar la familia de círculos:

FAMILIA DE CIRCULOS



▼ Solución P-5.1

Ecuación general de la circunferencia: centro (a,b) y radio c

```
ec = (x - a) ^2 + (y - b) ^2 == c^2
```

$(-a + x)^2 + (-b + y)^2 == c^2$

Círculo1: con centro en el eje OY ((a,b)=(0,b), a=0 y c=b)

```
ec1 = ec /. {a → 0, c → b}
```

$x^2 + (-b + y)^2 == b^2$

```
polar1 = ec1 /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

$r[t]^2 == 2\, b\, r[t]\, \text{Sin}[t]$

```
r[t]^2 == 2 b r[t] Sin[t]
```

$r[t]^2 == 2\, b\, r[t]\, \text{Sin}[t]$

```
Solve[polar1, r[t]]
```

{{r[t] → 0}, {r[t] → 2 b Sin[t]}}

```
circulo1[t_, b_] = 2 * b Sin[t];
```

Círculo2: con centro en el eje OX; tal que (a,b)=(a,0), b=0 y c=a

```
ec2 = ec /. {b → 0, c → a}
```

$(-a + x)^2 + y^2 == a^2$

```
polar2 = ec2 /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

$2\, a\, \text{Cos}[t]\, r[t] == r[t]^2$

```
Solve[polar2, r[t]]
```

{{r[t] → 0}, {r[t] → 2 a Cos[t]}}

```
circulo2[t_, a_] = 2 * a Cos[t];
```

Círculo3 con centro en el origen ; tal que (a,b)=(0,0), a=0 y b=0

```
ec3 = ec /. {a → 0, b → 0}
```

$x^2 + y^2 == c^2$

```
polar3 = ec3 /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

$c^2 == r[t]^2$

```
Solve[polar3, r[t]]
```

{{r[t] → -c}, {r[t] → c}}

```
circulo3[t_, a_] = a;
```

Familia de círculos

```
cir1 = PolarPlot[Evaluate[Table[circulo1[t, b], {b, 1, 5}]],
   {t, 0, 2 * π}, PlotStyle → RGBColor[1, 0, 0.2], PlotLabel → r == 2 b Sin[t]];
cir2 = PolarPlot[Evaluate[Table[circulo2[t, a], {a, 1, 5}]], {t, 0, π},
   PlotStyle → RGBColor[0.5, 1, 0], PlotLabel → r == 2 a Cos[t]];
cir3 = PolarPlot[Evaluate[Table[circulo3[t, c], {c, 2, 10, 2}]],
   {t, 0, 2 * π}, PlotStyle → RGBColor[0.5, 0, 1], PlotLabel → r == c ];
Show[cir1, cir2, cir3, PlotLabel → Style["FAMILIA DE CIRCULOS",
   Bold, 14, RGBColor[0.3, 0.2, 0.7]]]
```



▼ Ejercicio Propuesto P-5.2

Obtener la gráfica de las familias de cardiodes:



▼ Solución P-5.2

Cardioide 1

```
cardioide1[t_, a_] = a (1 + Cos[t]);
car1 = PolarPlot[cardioide1[t, 1], {t, 0, 2 π}, PlotStyle → Red]
```



Cardioide 2

```
cardioide2[t_, a_] = a (1 - Cos[t]);
car2 = PolarPlot[cardioide2[t, 1], {t, 0, 2 π}, PlotStyle → Green]
```



Cardioide 3

```
cardioide3[t_, a_] = a (1 + Sin[t]);
car3 = PolarPlot[cardioide3[t, 1], {t, 0, 2 π}, PlotStyle → Purple]
```



Cardioide 4

```
cardioide4[t_, a_] = a (1 - Sin[t]);
car4 = PolarPlot[cardioide4[t, 1], {t, 0, 2 π}, PlotStyle → Orange]
```



Familia de cardioides

```
Show[car1, car2, car3, car4]
```



Ejercicio Propuesto P-5.3

Obtener la gráfica de las familia de lemniscatas:



Solución P-5.3

```
lemniscata[t_, a_] = a (Cos[2 * t]) ^ (1 / 2);

PolarPlot[Evaluate[Table[lemniscata[t, a], {a, 2, 8, 2}]], {t, 0, 2 Pi},
 PlotLabel → Style["FAMILIA DE LEMNISCATAS", Bold, 14],
 Background → LightGreen]
```



Ejercicio Propuesto P-5.4

Obtener la gráfica de las espirales:



Solución P-5.4

```
esparq[θ_, a_, b_, x_] = a + b θ^ (1 / x);
```

```
PolarPlot[{esparq[θ, 0, 1, 1], esparq[θ, 0, -1, 1]},
 {θ, 0, 4 Pi}, PlotStyle → {Purple, Orange}]
```